////////////////////////////////////////////////////////////////////////////////
// Copyright 2010 Michael Schmalle - Teoti Graphix, LLC
// 
// Licensed under the Apache License, Version 2.0 (the "License");
// you may not use this file except in compliance with the License.
// You may obtain a copy of the License at
// 
// http://www.apache.org/licenses/LICENSE-2.0 
// 
// Unless required by applicable law or agreed to in writing, software 
// distributed under the License is distributed on an "AS IS" BASIS, 
// WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
// See the License for the specific language governing permissions and 
// limitations under the License
// 
// Author: Michael Schmalle, Principal Architect
// mschmalle at teotigraphix dot com
////////////////////////////////////////////////////////////////////////////////

package org.teotigraphix.swizplugin.events
{

import flash.events.Event;

import org.teotigraphix.swizplugin.registry.IExtension;
import org.teotigraphix.swizplugin.registry.IExtensionDescriptor;
import org.teotigraphix.swizplugin.registry.IExtensionPoint;

/**
 * The <code>ExtensionRegistry</code> change event.
 * 
 * @author Michael Schmalle
 * @copyright Teoti Graphix, LLC
 * @productversion 1.0
 */
public class ExtensionRegistryChangeEvent extends Event
{
	//--------------------------------------------------------------------------
	//
	//  Public :: Constants
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @eventType registryChange
	 */
	public static const REGISTRY_CHANGE:String = "registryChange";
	
	//--------------------------------------------------------------------------
	//
	//  Public :: Varaibles
	//
	//--------------------------------------------------------------------------
	
	/**
	 * The event kind, <code>ExtensionRegistryChangeKind</code> value.
	 */
	public var kind:String;
	
	/**
	 * The event element type, <code>RegistryChangeType</code> value.
	 */
	public var elementType:int;
	
	/**
	 * The instance that is the target of the current operation.
	 */
	public var element:Object;
	
	//--------------------------------------------------------------------------
	//
	//  Public :: Properties
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Returns the typed <code>IExtensionPoint</code> if 
	 * <code>elementType</code> is <code>RegistryClientType#EXTENSION_POINT</code>.
	 */
	public function get extensionPoint():IExtensionPoint
	{
		return element as IExtensionPoint;
	}
	
	/**
	 * Returns the typed <code>IExtension</code> if <code>elementType</code> is 
	 * <code>RegistryClientType#EXTENSION</code>.
	 */
	public function get extension():IExtension
	{
		return element as IExtension;
	}
	
	/**
	 * Returns the typed <code>IExtensionDescriptor</code> if 
	 * <code>elementType</code> is <code>RegistryClientType#EXTENSION_DESCRIPTOR</code>.
	 */
	public function get extensionDescriptor():IExtensionDescriptor
	{
		return element as IExtensionDescriptor;
	}
	
	//--------------------------------------------------------------------------
	//
	//  Constructor
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Creates a new instance of the ExtensionRegistryChangeEvent.
	 * 
	 * <p>Note: This event is used by an <code>IExtensionRegistry</code>.</p>
	 * 
	 * @param kind The String kind of event.
	 * @param elementType The int element type.
	 * @param element The element that is target of the current operation.
	 */
	public function ExtensionRegistryChangeEvent(kind:String,
												 elementType:int,
												 element:Object)
	{
		super(REGISTRY_CHANGE, false, false);
		
		this.kind = kind;
		this.elementType = elementType;
		this.element = element;
	}
	
	//--------------------------------------------------------------------------
	//
	//  Overridden Public :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @inheritDoc
	 */
	public override function clone():Event
	{
		return new ExtensionRegistryChangeEvent(kind, elementType, element);
	}
	
	/**
	 * @private
	 */
	public static function getTypeString(type:int):String
	{
		switch (type)
		{
			case ExtensionRegistryChangeType.CONTRIBUTOR:
				return "CONTRIBUTOR";
			case ExtensionRegistryChangeType.EXTENSION:
				return "EXTENSION";
			case ExtensionRegistryChangeType.EXTENSION_POINT:
				return "EXTENSION_POINT";
			case ExtensionRegistryChangeType.EXTENSION_DESCRIPTOR:
				return "EXTENSION_DESCRIPTOR";
		}
		
		return "";
	}
}
}